
1. [bookmark: _Toc122434485][bookmark: _Toc72225287]Overview
This document lists all the essential changes needed to correct problems in the TTCN implementation of test case 11.4.4 which is part of the NR5GC test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 
Contact:	Pengchong Shen
	shenpengchong@datangmobile.cn
	                                 Shaun Harry
	                                    shaun.harry@keysight.com
0. [bookmark: _Toc72225288]Verification Test Summary
Test Case: 	11.4.4
ATS Version:	iwd-TTCN3-B2020-09_D21wk12
System Simulator used:	Datang LinkTester 5G Protocol Conformance Test System ECT9610
Keysight S8704A Protocol Conformance Toolset
UE used:	HiSilicon Balong 5000
Verification Status:	PASS

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc72225289]Corrections required
1. [bookmark: _Toc72225290][bookmark: OLE_LINK4][bookmark: OLE_LINK5][bookmark: OLE_LINK6]Change 1
	Function name
	fl_TC_11_4_4_TestBody

	Reason for change
	1. According to TS 24.501 cl 5.5.1.3.2 case k), the UE in state 5GMM-
REGISTERED.ATTEMPTING-REGISTRATION-UPDATE shall initiate the registration procedure for mobility and periodic registration update when the upper layers request to establish an emergency PDU session. The 5GS registration type IE in REGISTRATION REQUEST message is set to "mobility registration updating" .

Note: it’s changed with the CR R5-212434.

2. There is no SRB2 is established during the emergency registration procedure, the current TTCN is missed.

	Summary of change
	1. Modify the emergency to mobility at step8 and add SMC procedure.
2. Modify the function f_NR5GC_EmergencyCallPDUSession (change 3),  add new input parameter.

	TTCN module
	Emergency_NR5GC.ttcn

	MCC160 Comment
	


Before Change
	  function fl_TC_11_4_4_NR5GC_TestBody() runs on NR5GC_PTC
  {
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;
    var O1_Type v_PDUSessionIdToRelease;
    var PDUSessionInfoList_Type v_PDUSessionInfo;
    var integer i;
    var integer v_NoOfSessions;

    // @siclog "Step 1" siclog@
    // The SS configures:
    //  - NR Cell 11 as "Serving cell"
    //  - NR Cell 1 as "Non-Suitable "Off" cell".

    f_NR_SetCellPower (nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE);
    f_NR_SetCellPower (nr_Cell11, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2);
    
    // @siclog "Steps 2 - 4" siclog@
    // The UE transmits an RRCConnectionRequest message.
    // SS transmit an RRCConnectionSetup message.
    // The UE transmits an RRCConnectionSetupComplete message to confirm the successful completion of the connection establishment
    // and a REGISTRATION REQUEST message indicating "mobility registration updating" is sent to update the registration of the actual tracking area.
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (nr_Cell11, tsc_NR_RRC_TI_Def, ?, tsc_SHT_IntegrityProtected);
    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Mobility)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");
    }
    //@siclog "Step 5" siclog@
    // SS sends a REGISTRATION REJECT message containing 5GMM cause value = #22 (Congestion).
    SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell11,
                                   tsc_NR_RbId_SRB1, // No SRB2 so must be on SRB1
                                   cs_TimingInfo_Now,
                                   cs_NG_NAS_Request(tsc_SHT_NoSecurityProtection, cs_NG_REGISTRATION_REJECT(cs_GMM_GSM_Cause(omit, tsc_NR5GCCause_Congestion), cs_GPRS_Timer2_3_IEI('5F'O, '001'B, '00011'B)))));
    
    //@siclog "Step 6" siclog@
    // The SS transmits an RRCConnectionRelease message.
    f_NR_RRCRelease(nr_Cell11);
    
    //@siclog "Step 7" siclog@
    //Make the UE attempt an IMS emergency call dialling an emergency number e.g. 112 or 911). (NOTE 1)
    f_UT_RequestIMSEmergencyCall(UT, px_EmergencyCallNumber);
    
    //@siclog "Step 8" siclog@
    // Check: Does the UE perform the Registration procedure for mobility registration update as specified in TS 38.508-1 [4] subclause 4.9.5, 'connected without release'?
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (nr_Cell11, tsc_NR_RRC_TI_Def, ?);
    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Emergency)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");
    }
       
    f_NR5GC_MobilityRegistration_Steps4_6(nr_Cell11, v_GMM_MobilityInfo, -, -, noRrcConnectionRelease);
    
    // @siclog "Step 8A - 9D" siclog@
    //Check: Does the UE transmit an UL NAS TRANSPORT message with 'Request type' set to 'initial emergency request', and, a PDU SESSION ESTABLISHMENT REQUEST for establishing an emergency PDU?
    // Steps 9-13 from Generic Test Procedure for IMS Emergency call establishment with IMS Emergency registration as specified in TS 38.508-1 [4], subclause 4.9.11, including the parallel behaviour specified in table 4.9.11.2.2-2 are performed.
    //f_NR5GC_EmergencyCallWithRegistration(nr_Cell11);
    f_NR5GC_EmergencyCallPDUSession(nr_Cell11);
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS2]);
    // Emergency call now active
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 8A");
    
    // @siclog "Step 10 - 11" siclog@
    // Make the UE release the emergency call. (NOTE 1)
    f_NR5GC_EmergencyCallRelease(nr_Cell11);
    
    // Release IP handling as this will be (re)started when this PDU is (re)established in the next loop
    f_IP_Handling_StopPDN(IP, cs_IP_StopPDN(PDN_2));

    v_PDUSessionInfo := f_NR5GC_MobileInfo_GetSessionInfoList();
    v_NoOfSessions := f_NR5GC_MobileInfo_GetNoOfSessions();
    // @siclog "Step 12" siclog@
    for (i := 0; i<v_NoOfSessions; i:=i+1)
    {
        v_PDUSessionIdToRelease := v_PDUSessionInfo[i].SessionId;
        f_NR5GC_PDUSessionRelease(nr_Cell11, // @sic R5s201131 sic@
                                v_PDUSessionIdToRelease,
                                cs_GMM_GSM_Cause (omit, '00011010'B), //5GSM cause Insufficient resources
                                cs_GPRS_Timer2_3_IEI('37'O, '101'B, '00101'B)); //T3396 = 5 minutes //
    }
    // @siclog "Step 13" siclog@
    // Void
    
    // @siclog "Step 14" siclog@
    // The SS releases the RRC connection.
    f_NR_RRCRelease(nr_Cell11);
    
    //@siclog "Step 15" siclog@
    //Make the UE attempt an IMS emergency call dialling an emergency number e.g. 112 or 911). (NOTE 1)
    // Prose CR RRC Release
    fl_EmergencyCallAndRelease(nr_Cell11, px_EmergencyCallNumber, "Step 16");
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 16 ");
  }


After Change
	   function fl_TC_11_4_4_NR5GC_TestBody() runs on NR5GC_PTC
  {
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;
    var O1_Type v_PDUSessionIdToRelease;
    var PDUSessionInfoList_Type v_PDUSessionInfo;
    var integer i;
    var integer v_NoOfSessions;

f_SMT_Step_Log(__FILE__, __LINE__,"Step 1");    // @siclog "Step 1" siclog@
    // The SS configures:
    //  - NR Cell 11 as "Serving cell"
    //  - NR Cell 1 as "Non-Suitable "Off" cell".

    f_NR_SetCellPower (nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE);
    f_NR_SetCellPower (nr_Cell11, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2);
	    
    // @siclog "Steps 2 - 4" siclog@
    // The UE transmits an RRCConnectionRequest message.
    // SS transmit an RRCConnectionSetup message.
    // The UE transmits an RRCConnectionSetupComplete message to confirm the successful completion of the connection establishment
    // and a REGISTRATION REQUEST message indicating "mobility registration updating" is sent to update the registration of the actual tracking area.
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (nr_Cell11, tsc_NR_RRC_TI_Def, ?, tsc_SHT_IntegrityProtected);
    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Mobility)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");
    }
    //@siclog "Step 5" siclog@
    // SS sends a REGISTRATION REJECT message containing 5GMM cause value = #22 (Congestion).
    SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell11,
                                   tsc_NR_RbId_SRB1, // No SRB2 so must be on SRB1
                                   cs_TimingInfo_Now,
                                   cs_NG_NAS_Request(tsc_SHT_NoSecurityProtection, cs_NG_REGISTRATION_REJECT(cs_GMM_GSM_Cause(omit, tsc_NR5GCCause_Congestion), cs_GPRS_Timer2_3_IEI('5F'O, '001'B, '00011'B)))));
    
    //@siclog "Step 6" siclog@
    // The SS transmits an RRCConnectionRelease message.
    f_NR_RRCRelease(nr_Cell11);
    
    //@siclog "Step 7" siclog@
    //Make the UE attempt an IMS emergency call dialling an emergency number e.g. 112 or 911). (NOTE 1)
    f_UT_RequestIMSEmergencyCall(UT, px_EmergencyCallNumber);
    
    //@siclog "Step 8" siclog@
    // Check: Does the UE perform the Registration procedure for mobility registration update as specified in TS 38.508-1 [4] subclause 4.9.5, 'connected without release'?
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (nr_Cell11, tsc_NR_RRC_TI_Def, ?);
    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Mobility)) { 
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");
    }

    f_NR_RRC_ActivateSecurity (nr_Cell11, valueof(v_ReceivedMsg.SecurityProtection.NasCount)); 
       
    f_NR5GC_MobilityRegistration_Steps4_6(nr_Cell11, v_GMM_MobilityInfo, -, -, noRrcConnectionRelease);
    
    // @siclog "Step 8A - 9D" siclog@
    //Check: Does the UE transmit an UL NAS TRANSPORT message with 'Request type' set to 'initial emergency request', and, a PDU SESSION ESTABLISHMENT REQUEST for establishing an emergency PDU?
    // Steps 9-13 from Generic Test Procedure for IMS Emergency call establishment with IMS Emergency registration as specified in TS 38.508-1 [4], subclause 4.9.11, including the parallel behaviour specified in table 4.9.11.2.2-2 are performed.
    //f_NR5GC_EmergencyCallWithRegistration(nr_Cell11);
    f_NR5GC_EmergencyCallPDUSession(nr_Cell11 , tsc_NR_RbId_SRB1);
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS2]);
    // Emergency call now active
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 8A");
    
    // @siclog "Step 10 - 11" siclog@
    // Make the UE release the emergency call. (NOTE 1)
    f_NR5GC_EmergencyCallRelease(nr_Cell11);
    
    // Release IP handling as this will be (re)started when this PDU is (re)established in the next loop
    f_IP_Handling_StopPDN(IP, cs_IP_StopPDN(PDN_2));

    v_PDUSessionInfo := f_NR5GC_MobileInfo_GetSessionInfoList();
    v_NoOfSessions := f_NR5GC_MobileInfo_GetNoOfSessions();
    // @siclog "Step 12" siclog@
    for (i := 0; i<v_NoOfSessions; i:=i+1)
    {
        v_PDUSessionIdToRelease := v_PDUSessionInfo[i].SessionId;
        f_NR5GC_PDUSessionRelease(nr_Cell11, // @sic R5s201131 sic@
                                v_PDUSessionIdToRelease,
                                cs_GMM_GSM_Cause (omit, '00011010'B), //5GSM cause Insufficient resources
                                cs_GPRS_Timer2_3_IEI('37'O, '101'B, '00101'B)); //T3396 = 5 minutes 
    }
    // @siclog "Step 13" siclog@
    // Void
    
    // @siclog "Step 14" siclog@
    // The SS releases the RRC connection.
    f_NR_RRCRelease(nr_Cell11);
    
    //@siclog "Step 15" siclog@
    //Make the UE attempt an IMS emergency call dialling an emergency number e.g. 112 or 911). (NOTE 1)
    // Prose CR RRC Release
    fl_EmergencyCallAndRelease(nr_Cell11, px_EmergencyCallNumber, "Step 16");
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 16 ");

  }


1. [bookmark: _Toc72225291][bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665]Change 2
	Function name
	f_NR5GC_PDUSessionEstablishment

	Reason for change
	1. For emergency PDU session establishment, the delay for user place signalling should be skipped. Because emergency IMS call will be setup immediately following emergency registration, the RRC connection will not be released anyway and the trigger from IMS PTC to establish additional QoS flow can come while the timer is running.

	Summary of change
	1.  Add a parameter p_IsEmergency and skip f_NR_DelayForUserPlaneSignalling() in emergency case

	TTCN module
	NR5GC_NASSteps.ttcn

	MCC160 Comment
	


Before Change
	  function f_NR5GC_PDUSessionEstablishment(NR_CellId_Type p_NR_CellId,
                                           integer p_ExpectedNumberOfNewPDUSessions := pc_noOf_PDUsSameConnection,  // @sic R5s201358 sic@,
                                           NR_RadioBearerId_Type p_SRB := tsc_NR_RbId_SRB2,
                                           template (omit) SDAP_Config p_SDAP_Config := omit,
                                           template S_NSSAI_Type p_ExpectedS_NSSAI := *,
                                           template DNN p_ExpectedDNN := *,
                                           template (value) SSC_Mode p_SSC_Mode := crs_SSC_Mode(-, '001'B),
                                           template (omit) DNN p_DNN := omit,
                                           template (omit) QoS_Rules p_QoS_Rules := omit,
                                           template (value) Session_AMBR p_Session_AMBR := cs_Session_AMBR ('06'O, '05'O, '0004'O, '05'O, '0004'O),
                                           template (omit) S_NSSAI_Type p_S_NSSAI := cs_S_NSSAI_SST1eMBB_WithIEI, // @sic R5s190919 sic@
                                           template (omit) QoSFlowDescr p_QoSFlowDescr := omit,
                                           template (present) O1_Type p_PDU_SessionId := ?,
                                           template ProcedureTransactionIdentifier p_PTI := omit,
                                           template (omit) PDU_SessionType p_PDU_SessionType := omit,
                                           template SSC_Mode p_ExpectedSSC_Mode := *, // @sic R5s190290 sic@
                                           template NG_UE_SM_Cap p_UECap := *,
                                           template MaxNumPacketFilters p_MaxNumPacketFilters := *,
                                           template IntegrityProtMaxDataRate p_IntegrityProtMaxDataRate := ?,
                                           template AlwaysOnPDUSessionReq p_AlwaysOnPDUSessionReq := *,
                                           template SM_PDU_DN_RequestContainer p_PDUReq := *,
                                           template ExtdProtocolConfigOptions p_ExtdPCO := *,
                                           template NG_PDU_SessionId p_OldPDU_SessionId := *,
                                           template NG_Request_Type p_RequestType := cr_NG_Request_Type('001'B),
                                           template AdditionalInformation p_AdditionalInfo := *,
                                           template (omit) GMM_GSM_Cause p_Cause := omit,
                                           template (omit) GPRS_Timer p_RQTimer := omit,
                                           template (omit) EAP_Message p_EAP := omit,
                                           template (omit) GPRS_Timer3 p_BackOff := omit,
                                           boolean p_PDUSessionAuthenticationRequired := false,
                                           template (omit) charstring p_TriggerWithThisAPN := omit,
                                           boolean p_FailWhenK_LessN := true,
                                           template (omit) charstring p_BuildQosFlow := omit,
                                           template (omit) EUTRA_NR_PduSessionInfoList_Type p_PduInfoFromLTE := omit) runs on NR5GC_PTC // @sic R5-198184, R5-199075 sic@
  {

……

    // Wait for IP signalling @sic R5-204329, R5-206296 sic@
    if (v_IMS_Configured) {
      v_WaitForUserplaneSignalling := waitForIMS;
    }  else { // currently all PDUs require IP signalling
      v_WaitForUserplaneSignalling := dontWaitForIMS;
    }
    f_NR_DelayForUserPlaneSignalling(v_WaitForUserplaneSignalling);  // @sic R5-206296 sic@

    // all expected msgs have been processed, so deactivate the extra default
    deactivate(v_MyDefaultVar);
  }


After Change
	  function f_NR5GC_PDUSessionEstablishment(NR_CellId_Type p_NR_CellId,
                                           integer p_ExpectedNumberOfNewPDUSessions := pc_noOf_PDUsSameConnection,  // @sic R5s201358 sic@,
                                           NR_RadioBearerId_Type p_SRB := tsc_NR_RbId_SRB2,
                                           template (omit) SDAP_Config p_SDAP_Config := omit,
                                           template S_NSSAI_Type p_ExpectedS_NSSAI := *,
                                           template DNN p_ExpectedDNN := *,
                                           template (value) SSC_Mode p_SSC_Mode := crs_SSC_Mode(-, '001'B),
                                           template (omit) DNN p_DNN := omit,
                                           template (omit) QoS_Rules p_QoS_Rules := omit,
                                           template (value) Session_AMBR p_Session_AMBR := cs_Session_AMBR ('06'O, '05'O, '0004'O, '05'O, '0004'O),
                                           template (omit) S_NSSAI_Type p_S_NSSAI := cs_S_NSSAI_SST1eMBB_WithIEI, // @sic R5s190919 sic@
                                           template (omit) QoSFlowDescr p_QoSFlowDescr := omit,
                                           template (present) O1_Type p_PDU_SessionId := ?,
                                           template ProcedureTransactionIdentifier p_PTI := omit,
                                           template (omit) PDU_SessionType p_PDU_SessionType := omit,
                                           template SSC_Mode p_ExpectedSSC_Mode := *, // @sic R5s190290 sic@
                                           template NG_UE_SM_Cap p_UECap := *,
                                           template MaxNumPacketFilters p_MaxNumPacketFilters := *,
                                           template IntegrityProtMaxDataRate p_IntegrityProtMaxDataRate := ?,
                                           template AlwaysOnPDUSessionReq p_AlwaysOnPDUSessionReq := *,
                                           template SM_PDU_DN_RequestContainer p_PDUReq := *,
                                           template ExtdProtocolConfigOptions p_ExtdPCO := *,
                                           template NG_PDU_SessionId p_OldPDU_SessionId := *,
                                           template NG_Request_Type p_RequestType := cr_NG_Request_Type('001'B),
                                           template AdditionalInformation p_AdditionalInfo := *,
                                           template (omit) GMM_GSM_Cause p_Cause := omit,
                                           template (omit) GPRS_Timer p_RQTimer := omit,
                                           template (omit) EAP_Message p_EAP := omit,
                                           template (omit) GPRS_Timer3 p_BackOff := omit,
                                           boolean p_PDUSessionAuthenticationRequired := false,
                                           template (omit) charstring p_TriggerWithThisAPN := omit,
                                           boolean p_FailWhenK_LessN := true,
                                           template (omit) charstring p_BuildQosFlow := omit,
                                           template (omit) EUTRA_NR_PduSessionInfoList_Type p_PduInfoFromLTE := omit,
                                           boolean p_IsEmergency := false) runs on NR5GC_PTC // @sic R5-198184, R5-199075 sic@
  {

……

if(not p_IsEmergency) {  
    // Wait for IP signalling @sic R5-204329, R5-206296 sic@
    if (v_IMS_Configured) {
      v_WaitForUserplaneSignalling := waitForIMS;
    }  else { // currently all PDUs require IP signalling
      v_WaitForUserplaneSignalling := dontWaitForIMS;
    }
    f_NR_DelayForUserPlaneSignalling(v_WaitForUserplaneSignalling);  // @sic R5-206296 sic@
    }
    // all expected msgs have been processed, so deactivate the extra default
    deactivate(v_MyDefaultVar);
  }


1. [bookmark: _Toc72225292]Change 3
	Function name
	f_NR5GC_EmergencyCallPDUSession

	Reason for change
	1. For change 1 point 2 ,there is no SRB2 is established during the emergency registration procedure, the current TTCN is missed.
2. Support change 2.2 emergency PDU session establishment, the delay for user place signalling should be skipped.

	Summary of change
	1. Add input parameter p_SRB.
2. Modify the parameter p_IsEmergency to true.

	TTCN module
	NR5GC_IMSProcedures.ttcn

	MCC160 Comment
	



Before Change
	  function f_NR5GC_EmergencyCallPDUSession(NR_CellId_Type p_NR_CellId) runs on NR5GC_PTC
  {
    var template (omit) GSM_MobilityInfo_Type v_GSM_MobilityInfo;

    f_NR5GC_PDUSessionEstablishment(p_NR_CellId,
                                    1,
                                    -, -,
                                    omit,
                                    omit,
                                    crs_SSC_Mode('A'H, '001'B),
                                    -,
                                    f_BuildDefaultQoSRules("2"),
                                    -, -, -, -, -, -, -, -, -, -, -, -, -, -,
                                    cr_NG_Request_Type('011'B),
                                    -, -, -, -, -, -, -, -,
                                    "2");

    // Get the Info for the PDU Session just created
    v_GSM_MobilityInfo := f_Get_PDUSessionForDNNType(f_NR5GC_MobileInfo_GetSessionInfoList(), Emergency_PDN);
    
    // Wait until need to add voice QoS
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS2]);
    f_NR5GC_ModifyPDUSession_AddVoice(p_NR_CellId, valueof(v_GSM_MobilityInfo));
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS2]);
  }



After Change
	 function f_NR5GC_EmergencyCallPDUSession(NR_CellId_Type p_NR_CellId,
										  NR_RadioBearerId_Type p_SRB := tsc_NR_RbId_SRB2 ) runs on NR5GC_PTC 
  {
    var template (omit) GSM_MobilityInfo_Type v_GSM_MobilityInfo;

    f_NR5GC_PDUSessionEstablishment (p_NR_CellId,
                                              1,
                                              p_SRB, -, 
                                              omit,
                                              omit,
                                              crs_SSC_Mode('A'H, '001'B),
                                              -,
                                              f_BuildDefaultQoSRules("2"),
                                              -, -, -, -, -, -, -, -, -, -, -, -, -, -,
                                              cr_NG_Request_Type('011'B),
                                              -, -, -, -, -, -, -, -,
                                              "2",
                                               -, true);

    // Get the Info for the PDU Session just created
    v_GSM_MobilityInfo := f_Get_PDUSessionForDNNType(f_NR5GC_MobileInfo_GetSessionInfoList(), Emergency_PDN);
    
    // Wait until need to add voice QoS
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS2]);	    
    f_NR5GC_ModifyPDUSession_AddVoice(p_NR_CellId, valueof(v_GSM_MobilityInfo));
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS2]);
  }


1. [bookmark: _Toc72225293]Change 4
	Function name
	f_NR5GC_PDUSessionRelease

	Reason for change
	1. For step 12, it is required to release all active PDU sessions. The currently active PDU sessions will have no active DRBs associated with them as the DRBs were released at step 6 and not re-established. This can cause problems during f_NR5GC_PDUSessionRelease() as UE and SS can be configured to release non-active DRBs.

	Summary of change
	1.  Cross-check DRBs associated with each PDU session against the list of active DRBs. If there is no active DRB, then no DRB release information will be included in RRCReconfiguration.

Note : With this implementation, RRCReconfiguration message will not have any parameters to reconfigure at the UE. In this case (release PDU session while DRB is not active) it is probably more correct to send PDU SESSION RELEASE COMMAND in DlInformationtransfer.

	TTCN module
	NR5GC_NASSteps.ttcn

	MCC160 Comment
	


Before Change
	  function f_NR5GC_PDUSessionRelease (NR_CellId_Type p_NR_CellId,
                                      O1_Type p_PDU_SessionId,
                                      template (value) GMM_GSM_Cause p_Cause := cs_GMM_GSM_Cause (omit, '00011010'B),
                                      template (omit) GPRS_Timer3 p_BackOff := omit,
                                      template (omit) NGSM_CongestionReattemptInd p_NGSM_CongestionReattemptInd := omit, // @sic R5s190543 sic@,
                                      template (omit) ExtdProtocolConfigOptions p_ExtdPCO := omit,
                                      template (omit) EAP_Message p_EAP := omit,
                                      ProcedureTransactionIdentifier p_PTI := '00'O,
                                      boolean p_ReceiveComplete := true)  runs on NR5GC_PTC
  {
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var IntegerList_Type v_DRBIdList := f_NR_GetDRBList_ForGivenPDUSession(oct2int(p_PDU_SessionId));
    var NR_DRBInfo_Type v_DRBInfo;
    var template (value) RadioBearerConfig v_RadioBearerConfig;
    var template (value) CellGroupConfig v_CellGroupConfig;
    var template (value) NR_RLC_BearerToReleaseList_Type v_RLC_BearerToReleaseList;
    var template (value) DRB_ToReleaseList v_DRB_ToReleaseList;
    var template (value) NR_RadioBearerList_Type v_SS_Config;
    var template (value) NG_NAS_DL_Pdu_Type v_PiggyBackedMsgToSend;
    var PDUSessionInfoList_Type v_PDUSessionInfoList := f_NR5GC_MobileInfo_GetSessionInfoList();
    var integer i;
    
    // Build NAS command
    v_PiggyBackedMsgToSend := f_Get_NG_PDUSessionReleaseCommand(p_PDU_SessionId,
                                                                p_Cause,
                                                                p_BackOff,
                                                                p_NGSM_CongestionReattemptInd,
                                                                p_ExtdPCO,
                                                                p_EAP,
                                                                p_PTI);
    
    // Build RRC command for all the DRBs linked to this PDU Session @sic R5-202550 sic@
    for (i := 0; i < lengthof(v_DRBIdList); i := i+1) {
        v_DRBInfo := valueof(f_NR_GetDRBInfo_FromId(v_DRBIdList[i]));
      v_DRB_ToReleaseList[i] := v_DRBInfo.DRB_Id;
      v_RLC_BearerToReleaseList[i] := v_DRBInfo.RLC_Config.logicalChannelIdentity;
      v_SS_Config[i] := cs_NR_SS_RadioBearer_Release(cs_NR_RadioBearerId_DRB(v_DRBInfo.DRB_Id));
    }

    v_RadioBearerConfig := cs_38508_RadioBearerConfigDef(omit, omit, v_DRB_ToReleaseList);
    v_CellGroupConfig := cs_38508_CellGroupConfig(f_NR_CellInfo_Get_CellGroupId(p_NR_CellId), omit, v_RLC_BearerToReleaseList);  //@sic R5s201387 sic@

    // The SS transmits an RRCReconfiguration message with a drb-ToReleaseList and PDU SESSION RELEASE COMMAND
    f_NR_RRCReconfigNewDRBWithSSConfig(p_NR_CellId,
                                       v_RadioBearerConfig,
                                       v_SS_Config,
                                       v_CellGroupConfig,
                                       { cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_PiggyBackedMsgToSend) },
                                       tsc_NR_RbId_SRB1, -, false);
    
    if (p_ReceiveComplete) {
      // @sic R5-206266 sic@
      interleave{
      // The UE transmit an RRCReconfigurationComplete message
      []SRB.receive(car_NR_SRB_RrcPdu_IND(p_NR_CellId, tsc_NR_RbId_SRB1, cr_38508_RRCReconfigurationComplete));

      // The UE transmits an ULInformationTransfer message and a UL NAS TRANSPORT containing PDU SESSION RELEASE COMPLETE.
      []SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId,
                                            tsc_NR_RbId_SRB2,
                                            cr_NG_NAS_IndWithPiggybacking((tsc_SHT_IntegrityProtected_Ciphered,tsc_SHT_IntegrityProtected)))) -> value v_ReceivedAsp;
       }//KS@10.1.3.2
      if (not f_Check_NG_PDUSessionReleaseComplete (v_PDUSessionInfoList, v_ReceivedAsp.Signalling.Nas[0].Pdu, p_PDU_SessionId)) {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Release Complete failed");
      }
    }
    f_NR5GC_MobileInfo_SetSessionInfoList (v_PDUSessionInfoList); // @sic R5-206266 sic@
  }


After Change
	 function f_NR5GC_PDUSessionRelease (NR_CellId_Type p_NR_CellId,
                                      O1_Type p_PDU_SessionId,
                                      template (value) GMM_GSM_Cause p_Cause := cs_GMM_GSM_Cause (omit, '00011010'B),
                                      template (omit) GPRS_Timer3 p_BackOff := omit,
                                      template (omit) NGSM_CongestionReattemptInd p_NGSM_CongestionReattemptInd := omit, // @sic R5s190543 sic@,
                                      template (omit) ExtdProtocolConfigOptions p_ExtdPCO := omit,
                                      template (omit) EAP_Message p_EAP := omit,
                                      ProcedureTransactionIdentifier p_PTI := '00'O,
                                      boolean p_ReceiveComplete := true)  runs on NR5GC_PTC
  {
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var IntegerList_Type v_DRBIdList := f_NR_GetDRBList_ForGivenPDUSession(oct2int(p_PDU_SessionId));
    var IntegerList_Type v_ActivDrbIdList := f_NR_GetActiveDRBs(p_NR_CellId); 
var template (omit) NR_DRBInfo_Type v_DRBInfo;
    var template (value) RadioBearerConfig v_RadioBearerConfig;
    var template (value) CellGroupConfig v_CellGroupConfig;
	  var template (omit) NR_RLC_BearerToReleaseList_Type v_RLC_BearerToReleaseList := omit;
	  var template (omit) DRB_ToReleaseList v_DRB_ToReleaseList := omit;
    var template (value) NR_RadioBearerList_Type v_SS_Config := {}; 
    var template (value) NG_NAS_DL_Pdu_Type v_PiggyBackedMsgToSend;
    var PDUSessionInfoList_Type v_PDUSessionInfoList := f_NR5GC_MobileInfo_GetSessionInfoList();
    var integer i;
    var integer j;
    var integer k :=0;
    var boolean find;
    
    // Build NAS command
    v_PiggyBackedMsgToSend := f_Get_NG_PDUSessionReleaseCommand(p_PDU_SessionId,
                                                                p_Cause,
                                                                p_BackOff,
                                                                p_NGSM_CongestionReattemptInd,
                                                                p_ExtdPCO,
                                                                p_EAP,
                                                                p_PTI);
    
    // Build RRC command for all the DRBs linked to this PDU Session @sic R5-202550 sic@
    for (i := 0; i < lengthof(v_DRBIdList); i := i+1) {
     find := false;
      for(j := 0; j < lengthof(v_ActivDrbIdList); j := j+1) {
        if(v_DRBIdList[i] == v_ActivDrbIdList[j]) {
          find := true;
          break;
        }
      }
      if(find) {
        v_DRBInfo := valueof(f_NR_GetDRBInfo_FromId(v_DRBIdList[i]));
        v_DRB_ToReleaseList[k] := v_DRBInfo.DRB_Id;
        v_RLC_BearerToReleaseList[k] := v_DRBInfo.RLC_Config.logicalChannelIdentity;
        v_SS_Config[k] := cs_NR_SS_RadioBearer_Release(cs_NR_RadioBearerId_DRB(v_DRBInfo.DRB_Id));
        k := k+1;
      }
    }

    v_RadioBearerConfig := cs_38508_RadioBearerConfigDef(omit, omit, v_DRB_ToReleaseList);
    v_CellGroupConfig := cs_38508_CellGroupConfig(f_NR_CellInfo_Get_CellGroupId(p_NR_CellId), omit, v_RLC_BearerToReleaseList);  //@sic R5s201387 sic@

    // The SS transmits an RRCReconfiguration message with a drb-ToReleaseList and PDU SESSION RELEASE COMMAND
    f_NR_RRCReconfigNewDRBWithSSConfig(p_NR_CellId,
                                       v_RadioBearerConfig,
                                       v_SS_Config,
                                       v_CellGroupConfig,
                                       { cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_PiggyBackedMsgToSend) },
                                       tsc_NR_RbId_SRB1, -, false);
    
    if (p_ReceiveComplete) {
      // @sic R5-206266 sic@
      interleave{
      // The UE transmit an RRCReconfigurationComplete message
      []SRB.receive(car_NR_SRB_RrcPdu_IND(p_NR_CellId, tsc_NR_RbId_SRB1, cr_38508_RRCReconfigurationComplete));

      // The UE transmits an ULInformationTransfer message and a UL NAS TRANSPORT containing PDU SESSION RELEASE COMPLETE.
      []SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId,
                                            tsc_NR_RbId_SRB2,
                                            cr_NG_NAS_IndWithPiggybacking((tsc_SHT_IntegrityProtected_Ciphered,tsc_SHT_IntegrityProtected)))) -> value v_ReceivedAsp;
       }//KS@10.1.3.2
      if (not f_Check_NG_PDUSessionReleaseComplete (v_PDUSessionInfoList, v_ReceivedAsp.Signalling.Nas[0].Pdu, p_PDU_SessionId)) {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Release Complete failed");
      }
    }
    f_NR5GC_MobileInfo_SetSessionInfoList (v_PDUSessionInfoList); // @sic R5-206266 sic@
  }


1. [bookmark: _Toc72225294]Change 5
	Function name
	f_ReconfigureDRBs

	Reason for change
	1. For step 12, it is required to release all active PDU sessions. The currently active PDU sessions will have no active DRBs associated with them as the DRBs were released at step 6 and not re-established. This can cause problems during f_NR5GC_PDUSessionRelease() as UE and SS can be configured to release non-active DRBs.

	Summary of change
	1.  Check the size of p_SS_Config and skip the SS configuration if the list is empty

	TTCN module
	NR5GC_RRCSteps.ttcn

	MCC160 Comment
	



Before Change
	  function f_ReconfigureDRBs(NR_CellId_Type p_NR_CellId,
                             template (omit) NG_NAS_MSG_RequestList_Type p_NASMsg,
                             template (value) RadioBearerConfig p_RadioBearerConfig,
                             template (value) CellGroupConfig p_CellGroupConfig,
                             template (omit) NR_RadioBearerList_Type p_SS_Config,
                             NR_RadioBearerId_Type p_SRB := tsc_NR_RbId_SRB1,
                             template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,
                             boolean p_ReceiveForRRCreconfigurationComplete := true,
                             template (omit) NR_RoutingInfoSUL_Type p_NR_RoutingInfoSUL := omit,
                             template (omit) PDSCH_ServingCellConfig.nrofHARQ_ProcessesForPDSCH  p_NrofHARQ_ProcessesForPDSCH := omit,
                             template (omit) NR_SecurityParams_Type p_SecurityParams := omit) runs on NR_BASE_PTC
  { // Split function so list of DRBs can be built elsewhere @sic R5-199103 sic@
    // @sic R5-203533: added p_NrofHARQ_ProcessesForPDSCH sic@
    var template (value) RRCReconfiguration_v1530_IEs v_V1530Ext;
    var NR_SecurityParams_Type v_SecurityParams := f_NR_Security_Get();
    
    if (isvalue(p_NrofHARQ_ProcessesForPDSCH)) {
      p_CellGroupConfig.spCellConfig.spCellConfigDedicated.pdsch_ServingCellConfig := cs_NR_Pdsch_ServingCellConfigSetup(cs_NR_PDSCH_ServingCellConfigDef(-, -, p_NrofHARQ_ProcessesForPDSCH));
      f_NR_SS_CommonCellConfig(p_NR_CellId, cads_NR_PDSCH_Config_REQ(p_NR_CellId, p_TimingInfo, cs_NR_PDSCH_CellLevelConfigDef(p_CellGroupConfig.spCellConfig.spCellConfigDedicated.pdsch_ServingCellConfig.setup)));
    }
    v_V1530Ext := cs_NR_RRCReconfiguration_v1530_IEs(bit2oct(encvalue(p_CellGroupConfig)));

    //Reconfigure DRB in SS - existing DRBs are not reconfigured during release
      f_NR_SS_CommonRadioBearerConfig(p_NR_CellId, p_SS_Config, p_TimingInfo);
    
    // Configure PDCP security
    if (isvalue(p_SecurityParams)){ // @sic R5s210277 sic@
      v_SecurityParams := valueof(p_SecurityParams);
    } else {
      v_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetCurrent(p_NR_CellId, false);  // @sic R5s190382, R5s210277 sic@
    }
    f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_NR_CellId,
                                          v_SecurityParams.AS_Integrity,
                                          v_SecurityParams.AS_Ciphering);
    // Send message to UE
    f_NR_SendRRCReconfiguration (p_NR_CellId, p_SRB, p_RadioBearerConfig, v_V1530Ext, p_NASMsg, -, p_TimingInfo, p_ReceiveForRRCreconfigurationComplete, p_NR_RoutingInfoSUL);
  }



After Change
	 function f_ReconfigureDRBs(NR_CellId_Type p_NR_CellId,
                             template (omit) NG_NAS_MSG_RequestList_Type p_NASMsg,
                             template (value) RadioBearerConfig p_RadioBearerConfig,
                             template (value) CellGroupConfig p_CellGroupConfig,
                             template (omit) NR_RadioBearerList_Type p_SS_Config,
                             NR_RadioBearerId_Type p_SRB := tsc_NR_RbId_SRB1,
                             template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,
                             boolean p_ReceiveForRRCreconfigurationComplete := true,
                             template (omit) NR_RoutingInfoSUL_Type p_NR_RoutingInfoSUL := omit,
                             template (omit) PDSCH_ServingCellConfig.nrofHARQ_ProcessesForPDSCH  p_NrofHARQ_ProcessesForPDSCH := omit,
                             template (omit) NR_SecurityParams_Type p_SecurityParams := omit) runs on NR_BASE_PTC
  { // Split function so list of DRBs can be built elsewhere @sic R5-199103 sic@
    // @sic R5-203533: added p_NrofHARQ_ProcessesForPDSCH sic@
    var template (value) RRCReconfiguration_v1530_IEs v_V1530Ext;
    var NR_SecurityParams_Type v_SecurityParams := f_NR_Security_Get();
    
    if (isvalue(p_NrofHARQ_ProcessesForPDSCH)) {
      p_CellGroupConfig.spCellConfig.spCellConfigDedicated.pdsch_ServingCellConfig := cs_NR_Pdsch_ServingCellConfigSetup(cs_NR_PDSCH_ServingCellConfigDef(-, -, p_NrofHARQ_ProcessesForPDSCH));
      f_NR_SS_CommonCellConfig(p_NR_CellId, cads_NR_PDSCH_Config_REQ(p_NR_CellId, p_TimingInfo, cs_NR_PDSCH_CellLevelConfigDef(p_CellGroupConfig.spCellConfig.spCellConfigDedicated.pdsch_ServingCellConfig.setup)));
    }
    v_V1530Ext := cs_NR_RRCReconfiguration_v1530_IEs(bit2oct(encvalue(p_CellGroupConfig)));

    //Reconfigure DRB in SS - existing DRBs are not reconfigured during release
    if(lengthof(p_SS_Config)>0) { 
      f_NR_SS_CommonRadioBearerConfig(p_NR_CellId, p_SS_Config, p_TimingInfo);
    }
    // Configure PDCP security
    if (isvalue(p_SecurityParams)){ // @sic R5s210277 sic@
      v_SecurityParams := valueof(p_SecurityParams);
    } else {
      v_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetCurrent(p_NR_CellId, false);  // @sic R5s190382, R5s210277 sic@
    }
    f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_NR_CellId,
                                          v_SecurityParams.AS_Integrity,
                                          v_SecurityParams.AS_Ciphering);
    // Send message to UE
    f_NR_SendRRCReconfiguration (p_NR_CellId, p_SRB, p_RadioBearerConfig, v_V1530Ext, p_NASMsg, -, p_TimingInfo, p_ReceiveForRRCreconfigurationComplete, p_NR_RoutingInfoSUL);
  }


1. [bookmark: _Toc72225295]Branches executed
NR Band 41, nea2, Nia2.
1. [bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666][bookmark: _Toc72225296]Execution Log Files
3. [bookmark: _Toc43191966][bookmark: _Toc54877395][bookmark: _Toc72225297]HiSilicon Balong 5000
The HiSilicon Balong 5000 passed this test case on Datang LinkTester 5G Protocol Conformance Test System ECT9610 and Keysight S8704A Protocol Conformance Toolset . The documentation below is enclosed as evidence of the successful test case run [1]:
· Test Case Execution log file:
Datang LinkTester
\ TC_11_4_4_NR5GC_N41.log
Keysight Technologies UK
\ TC_11_4_4_LOG.html
· PIXCIT settings used:
Datang LinkTester
\ Pics(Hisilicon-IRAT).xml
\ Pixit(TC_11_4_4_NR5GC).xml
Keysight Technologies UK
\ TC_11_4_4_PIXIT.xml

